Pass the password back This code simply receives a very basic 'password' (numbers, letters, nothing fancy). After you press enter, it will display it back. The only special key I'm handling is backspace (code 8 in ASCII).

uses crt;

var
    pass: string;
    key: char;
    len: integer;

begin

clrscr;

key := chr(0);
pass := chr(0);

write('Write the password:');

while key&lt;&gt;chr(13) do
begin
    key := readkey;
    len := length(pass);
    case ord(key) of
        8: begin
                if len &gt; 0 then
                begin
                    delete(pass, len, 1);
                end;
            end;
        else
            pass := concat(pass, key);
    end;
end;

writeln('Password: ', pass);

readln; //pauses the code

end.


You will probably notice that the program block is missing. It was on purpose, since Free Pascal Compiler has issues with it.

The code compiles / runs fine, to the best of my knowledge. Remember that this is simply and purely made for ASCII 7-bit (I think the extended set will work).

In terms of clarity, structure and indentation, what else can I improve?
Indentation: 


From the Coding Style recommendation, I am using 2-space indentation, which works for me. The guide suggests that you should avoid tabs as well:


  Do not use TAB characters (ASCII HT, 0x09). There is no standard default TAB setting, so the look of source files using TAB characters will depend on client settings. This may result in a chaotic view of source files. Align by space characters (also see Indentation). 

However, the same coding style says that you should not use space around operators, symbols...:


  Don't use spaces around operators, colons, parentheses etc. e.g. write p:=p+i; instead of p := p + i ;.


Since I am not using colored symbols, I've chosen to follow the "Guideline of the Lazarus 1.4.0 color preview":



As you can see, the color preview of the IDE uses spaces around symbols (:=, +, -, *, and so on). Whoever wrote this preview is a point out of curve like me.


Code:


Repeat is a posttest loop, and your code needs to run at least once, so why not a repeat..until loop.. In another words, you know the first key will always be different from #13, you hardcoded it. You can save a begin..end block, as well.
You do not need begin..end blocks for one line statements inside case conditions.


Structure:

For code reuse and exploration I would find it more practical to create a function that would only wait for and then return the password:

program HandlePassword;

uses
  crt;

function WaitForPassword: string;
var
  key: char;
  len: integer;
begin
  Result := Chr(0);
  key := Chr(0);
  repeat
    key := ReadKey;
    len := Length(Result);
    case Ord(key) of
      8: if len &gt; 0 then Delete(Result, len, 1);
      else Result := Result + key;
    end;
  until key = Chr(13);
end;

begin
  ClrScr; // clear screen
  Write('Write the password:');
  WriteLn(WaitForPassword);
  ReadLn; // pauses the code
end.